Supplement S1. FORTRAN code of program BipDesc
c PROGRAM BipDesc
c
C INPUT OPTIONS AS GIVEN BELOW IN THE PROGRAM CODE
C BIPLOT DATA FOR EACH ORDINATION ARE PRESENTED IN A MATRIX
C WITH HB+HV ROWS (OBJECTS AND THEN VARIABLES) AND M COLUMNS
C (DIMENSIONS)
C K ORDINATIONS OF THE SAME SIZE ARE PRESWENTED IN THE SAME
C INPUT FILE ONE AFTER THE OTHER
C EXAMPLE FOR TWO ORDINATIONS, FOUR OBJECTS AND THREE VARIABLES
C   AND TWO DIMENSIONS    
C
C -2.39166  1.53253   
C -.766830 -.570336   
C -.112184 -.797694      
C  1.69082  .622514       
C  .635545  1.60376  
C  .713606 -.952451  
C  .482390  9.039950E-02  
C
C  1.45142 -1.97646   
C  .420561 -.686949  
C  .305508 -.122395   
C  .959591 -6.987646E-02  
C -.879201 -.295676   
C -.247628  .660158  
C  .386290  .768501  
C
C PRINTED OUTPUT
C  EUCLIDEAN DISTANCE MATRIX FOR K ORDINATIONS
C  NOMRLAIZING CONSZTANT
C  MATRIX OF NORMALIZED AUCLIDEAN DISTANCE
C
C OUTPUT SAVED IN FILE 'Q.TXT'
C K ORDINATIONS BY HB+HV DESCRIPTORS MATRIX FOR FURTHER ANALYSIS
C
      ALLOCATABLE D(:,:,:), R(:,:), SUMM (:), EUC(:,:) 
      character*25 FN1
      INTEGER HB, HV
      write(*,*) ' ENTER FILENAME FOR INPUT DATA'  
      READ(*,'(A)') FN1    
      OPEN(1,FILE=fn1)
      WRITE(*,*) ' ENTER NO. OF DIMENSIONS'
      READ (*,*) M 
      WRITE(*,*) ' ENTER NO. OF ORDINATIONS'
      READ (*,*) K 
      WRITE(*,*) ' ENTER NO. OF OBJECTS'
      READ (*,*) HB 
      WRITE(*,*) ' ENTER NO. OF VARIABLES'
      READ (*,*) HV
      WRITE(*,*) ' ANGLE WEIGHTING OPTION'
      WRITE(*,*) ' 0-VECTOR LENGTH SQUARED'
      WRITE(*,*) ' 1-VECTOR LENGTH NOT SQUARED'
      WRITE(*,*) ' ENTER YOUR CHOICE'
      READ (*,*) IOPT
      N=HB+HV
      IST=HB*HV
      ALLOCATE (D(K,N,M),R(K,IST),SUMM(N),EUC(K,K))
C      
      DO 100 I=1,K
      DO 100 L=1,N
      READ(1,*) (D(I,L,J), J=1,M)
 100  CONTINUE     
C
      DO 200 I=1,K
      AMAXVAR=0
      AMAXOBJ=0
      DO 155 L=1,N
      SUMM(L)=0
      DO 150 J=1,M
 150  SUMM(L)=SUMM(L)+D(I,L,J)**2
      IF(L.LE.HB.AND.AMAXOBJ.LT.SUMM(L)) AMAXOBJ=SUMM(L)
      IF(L.GT.HB.AND.AMAXVAR.LT.SUMM(L)) AMAXVAR=SUMM(L)
 155  CONTINUE
      IF(IOPT.EQ.1) THEN
      AMAXOBJ=SQRT(AMAXOBJ)
      AMAXVAR=SQRT(AMAXVAR)
      ENDIF
C     WRITE(*,*) (SUMM(J), J=1,N)
C     write(*,*) AMAXOBJ, AMAXVAR
C ANG SEP  
      ICN=0
      DO 180 II=1,HB
      DO 180 IJ=HB+1,N
      A=0
      DO 175 J=1,M
 175  A=A+D(I,II,J)*D(I,IJ,J)
      ICN=ICN+1
C ANGLE      
      R(I,ICN)=A/SQRT(SUMM(II)*SUMM(IJ))   
      IF(IOPT.EQ.0) THEN   
C ANGLE WEIGHTED BY MEAN REL. SQUARED VECTOR LENGTH
      R(I,ICN)= R(I,ICN)  
     X  * ((SUMM(II)/AMAXOBJ)+(SUMM(IJ)/AMAXVAR))/2.0
      ELSE
C ANGLE WEIGHTED BY MEAN REL. VECTOR LENGTH
      R(I,ICN)= R(I,ICN)  
     X  * ((SQRT(SUMM(II))/AMAXOBJ)+(SQRT(SUMM(IJ))/AMAXVAR))/2.0
      ENDIF
 180  CONTINUE     
 200  CONTINUE
C OUTPUT
      OPEN(2,FILE='Q.TXT')
      WRITE(2,*) fn1
      WRITE(2,*) K, IST
      DO 300  I=1,K
 300  WRITE(2,*) (R(I,J), J=1,IST)
      CLOSE (2)
C DISTANCE CALCULATION
      CONS=2.0*SQRT(FLOAT(IST))
      WRITE(*,*)
      WRITE(*,*) ' EUCLIDEAN DISTANCES BETWEEN BIPLOTS '
      WRITE(*,*)
      DO 400 II=1,K
      DO 399 IJ=II,K
      EUC(II,IJ)=0
      IF(II.EQ.IJ) GO TO 399
      DO 388 J=1,IST
 388  EUC(II,IJ)=EUC(II,IJ)+(R(II,J)-R(IJ,J))**2
      EUC(II,IJ)=SQRT(EUC(II,IJ))
 399  EUC(IJ,II)=EUC(II,IJ)
 400  WRITE(*,*) II, (EUC(II,KL), KL=1,K)
      WRITE(*,*)
      WRITE(*,*) ' NORMALIZING CONSTANT = ', CONS
      WRITE(*,*)
      WRITE(*,*) ' NORMALIZED EUCLIDEAN DISTANCES BETWEEN BIPLOTS '
      WRITE(*,*)
      DO 500 II=1,K
      DO 499 IJ=1,K
 499  EUC(II,IJ)=EUC(II,IJ)/CONS
 500  WRITE(*,*) II, (EUC(II,KL), KL=1,K)
      WRITE(*,*)
      STOP
      END

